
Maple-assisted derivation of recurrence for A296393

There are  configurations for a  sub-array, but not all can arise.

 We encode these configurations as lists in the order Each of 

 is horizontally, vertically or diagonally adjacent to 4 other entries, and those (plus the 
entry itself) can't all be 1's.

goodconfig:= proc(x)
   x[2]+x[1]+x[3]+x[6]+x[7]<=4
   and x[3]+x[2]+x[4]+x[7]+x[8]<=4
   and x[4]+x[3]+x[5]+x[8]+x[9]<=4
   and x[7]+x[2]+x[3]+x[6]+x[8]<=4
   and x[8]+x[3]+x[4]+x[7]+x[9]<=4
   and x[9]+x[4]+x[5]+x[8]+x[10]<=4
end proc:
Configs:= select(goodconfig, [seq(convert(2^10+i,base,2)[1..10],
i=0..2^10-1)]):
nops(Configs);  

 
912

There are  allowed configurations.
Consider the  transition matrix  with entries if the first two rows of a  sub-
array could be in configuration  while the last two rows are in configuration , and  otherwise.  The 
following code computes it.
Compatible:= proc(i,j) local Xi,Xj,k;
 Xi:= Configs[i]; Xj:= Configs[j];
 if Xi[6..10] <> Xj[1..5] then return 0 fi;
 if Xj[1]=1 and not member(Xi[1]+Xi[2]+Xj[2]+Xj[6], {1,3}) then 
return 0 fi;
 if Xj[2]=1 and not member(Xi[2]+Xi[3]+Xj[1]+Xj[3]+Xj[6]+Xj[7], 
{1,3}) then return 0 fi;
 if Xj[3]=1 and not member(Xi[3]+Xi[4]+Xj[2]+Xj[4]+Xj[7]+Xj[8], 
{1,3}) then return 0 fi;
 if Xj[4]=1 and not member(Xi[4]+Xi[5]+Xj[3]+Xj[5]+Xj[8]+Xj[9], 
{1,3}) then return 0 fi;
 if Xj[5]=1 and not member(Xi[5]+Xj[4]+Xj[9]+Xj[10], {1,3}) then 
return 0 fi;
 1
end proc:
T:= Matrix(912,912,Compatible):

Thus for   where  is a column vector with  for configurations whose first 

row could be a top row,  otherwise, and similarly  has  for configurations whose second row could 

be a bottom row.  
u:= Vector(912,proc(i) local x; x:= Configs[i];
   if x[1]=1 and x[2]+x[6] <> 1 then return 0 fi;
   if x[2]=1 and not member(x[1]+x[3]+x[6]+x[7], {1,3}) then 
return 0 fi;
   if x[3]=1 and not member(x[2]+x[4]+x[7]+x[8], {1,3}) then 
return 0 fi;
   if x[4]=1 and not member(x[3]+x[5]+x[8]+x[9], {1,3}) then 
return 0 fi;
   if x[5]=1 and not member(x[4]+x[9]+x[10], {1,3}) then return 0
fi;
   1
end proc):
v:= Vector(912,proc(i) local x; x:= Configs[i];
   if x[6]=1 and not member(x[1]+x[2]+x[7],{1,3}) then return 0 
fi;
   if x[7]=1 and not member(x[2]+x[3]+x[6]+x[8],{1,3}) then 
return 0 fi;
   if x[8]=1 and not member(x[3]+x[4]+x[7]+x[9],{1,3}) then 
return 0 fi;
   if x[9]=1 and not member(x[4]+x[5]+x[8]+x[10],{1,3}) then 
return 0 fi;
   if x[10]=1 and x[5]+x[9]<> 1 then return 0 fi;
   1
 end proc) :

To check, here are the first few entries of our sequence (including , which doesn't really fit the pattern,

although it will work with the recurrence).  For future use, we pre-compute more  than we need.
Tv[0]:= v:
for n from 1 to 132 do Tv[n]:= T . Tv[n-1] od:
A:= [6, seq(u^%T . Tv[n],n=0..132)]:
A[1..20];

To find a recurrence, we look for a linear dependence in the vectors . If , our recurrence 

will be 

M:= Tv[0]:
for n from 1 do
  M:= <M|Tv[n]>;
  if LinearAlgebra:-Rank(M) < n+1 then break fi;
od:
n;

125
Thus there must be a linear recurrence of order 125, i.e. is a linear combination of 

NM:=LinearAlgebra:-NullSpace(M)[1]:
NM;

Now here is the recurrence formula. 
sort(add(NM[i]*a[k+i-1],i=1..126),[seq(a[k+i],i=0..125)])=0;

We check that it is valid even for .
seq(add(A[k+i]*NM[i],i=1..126),k=0..8);


